Ordnance Survey
Information

March 1983

Transverse Mercator Projection

Constants, Formulae and Methods

1.Introduction

a. This pamphlet replaces the HMSO booklet "Constants Formulae and Methods Used in the Transverse Mercator Projection" which is now out of print. It describes briefly the co-ordinate systems in use and brings together some useful formulae used for computation on the Transverse Mercator Projection and for transformations to and from spheroidal systems. The description of calculations are related to the use of modern programmable desk top calculators which have rendered the Ordnance Survey projection tables obsolete.

b. To increase the usefulness of the pamphlet the constants for the International spheroid and associated UTM projection zones 30 and 31 which cover the British Isles are also given. The formulae are presented in a slightly modified form from the previous publication. Terms which were included in order to control the size of intermediate values (powers of 10) or to return the results in seconds of arc (sin 1" and powers) are dispensed with. All constants which appeared in the formulae and which only applied to the national projection have been replaced by generalised terms.

c. Worked examples for each of the formulae with intermediate values are given in section 7.

d. To aid the implementation of the formulae on programmable machines a number of sub routines written in BASIC computer language are given in Appendix A. No attempt has been made to optimise the run time characteristics of these routines. It was considered more important for the formulae to be recognisable than that they should be efficient. The worked examples in section 7 have been computed using these routines.

2. Reference Systems

Spheroidal

a. The form of the Earth (Geoid) cannot be precisely defined mathematically. The best practical representation of the form of the geoid can be obtained using an ellipsoid of rotation (spheroid). However no single spheroid can adequately represent the complete Geoid; there are therefore a number from which to choose in order to get the best local fit. The National projection is based on the Airy spheroid which gives a very good fit in the region of the British Isles. The International spheroid is used extensively in Europe.

Transverse Mercator

b. In the simple Transverse Mercator Projection the surface of the spheroid chosen to represent the Earth is represented on a cylinder which touches the spheroid along a chosen meridian and which is then unwrapped. The scale is therefore correct along this central meridian and increases on either side of it.

c. In practice the scale is usually made too small on the central meridian in the ratio of $\frac{9996}{10000}$ approximately. The projection then becomes correct in scale on two lines nearly parallel with and on either side of the central meridian and about two thirds of the way between it and the edges of the projection. On the edges, the projection scale will have increased to approximately $\frac{10004}{10000}$ of the nominal figure. The change in scale is most conveniently done in practice by applying (*Fo*) to all dimensions of the spheroid before calculating the projection.

3. Symbols and Definitions

a = Major Semi-axis
b = Minor Semi-axis

$$e^2 = \frac{a^2 - b^2}{a^2}$$

$$n = \frac{a-b}{a+b}$$

ν = radius of curvature at latitude ϕ perpendicular to a meridian $= \dfrac{a}{(1 - e^2 \sin^2 \phi)^{1/2}}$

ρ = radius of curvature of a meridian at latitude ϕ

$$= \frac{a(1-e^2)}{(1-e^2 \sin^2 \phi)^{3/2}} = \frac{\nu(1-e^2)}{(1-e^2 \sin^2 \phi)}$$

$$\eta^2 = \frac{\nu}{\rho} - 1 \quad = \quad \frac{e^2 \cos^2 \phi}{1 - e^2}$$

ϕ = Latitude of a point
λ_p = Longitude of a point measured East (+) or West (−) of Greenwich.
ϕ' = Latitude of the foot of the perpendicular drawn from a point on the projection to the central meridian.
ϕ_0 = Latitude of True origin.
λ_0 = Longitude of True origin.
Eo = Grid Eastings of True origin.
No = Grid Northings of True origin.
E = Grid Eastings (metres)
N = Grid Northings (metres)
$y = E - Eo$ = "True" Easting
$x = N - No$ = "True" Northing
Fo = Scale factor on the central meridian.
F = Scale factor at a point. (This is usually called Local Scale Factor).
S = True distance between two points on the spheroid.
s = Straight-line distance between two points on the projection.
A = True meridional arc
M = Developed meridional arc = $A \times F_0$
t = Straight-line direction joining two points on the projection.
T = Direction on the projection of the projected geodesic or line of sight joining two points.
C = "Convergence" of meridians on the projection i.e. the angle at any point in the projection between the North-South Grid Line and the meridian at that point.
$P = \lambda_p - \lambda_0$

4. Constants

	National Projection	Universal Transverse Mercator Projection (UTM)	
Spheroid	Airy	International (1924)	
a	6377563·396 m	6378388·000 m * $f = \frac{1}{297}$	
b	6356256·910 m	6356911·946 m	
	* $f = 1/299.32498$		
True Origin		Zone 30 **	Zone 31
	Lat 49° N	Lat 0°	Lat 0°
	Long 2° W	Long 3° W	Long 3° E
False Origin	E 400 000 m W of True origin	E 500 000 m W of True origin	
	N 100 000 m N of True origin	N 0 m	
Grid Co-ordinates of True Origin			
E	400 000 m	500 000 m	
N	−100 000 m	0 m	
Scale on Central Meridian (Fo)	0·9996012717	0·9996	

** Long. of central meridian of Zone Z $= 6Z - 183$, $Z = 1, 2, \ldots, 60$

* f = Polar flattening = $\frac{a-b}{a}$

5. Useful Constants

	National projection	UTM
$a \times Fo$	6375020·481	6375836·645
$b \times Fo$	6353722·490	6354369.181
n	0·001673220250	0·001686340651
e^2	0·006670539762	0·006722670062

6. Formulae

The formulae given below assume that the linear quantities are already in international metres *and that they have been scaled by Fo*. All angles are expressed in radians, including ϕ and λ.

a. Developed Arc of a Meridian from ϕ_2 to ϕ_1

$$
\begin{aligned}
M\phi_2 - M\phi_1 = b\{&(1 + n + \tfrac{5}{4}n^2 + \tfrac{5}{4}n^3)(\phi_2 - \phi_1) \\
&- (3n + 3n^2 + \tfrac{21}{8}n^3)\sin(\phi_2 - \phi_1)\cos(\phi_2 + \phi_1) \\
&+ (\tfrac{15}{8}n^2 + \tfrac{15}{8}n^3)\sin 2(\phi_2 - \phi_1)\cos 2(\phi_2 + \phi_1) \\
&- \tfrac{35}{24}n^3 \sin 3(\phi_2 - \phi_1)\cos 3(\phi_2 + \phi_1)\}
\end{aligned}
$$

NB 1. M in paragraph (b) below is obtained by making ϕ_1 in this equation equal to ϕ_0.

2. ϕ in paragraph (c) and (e) below is obtained by iteration. As in (1) above ϕ_1 is set to ϕ_0 and M is computed using a trial value ϕ' for ϕ_2. If ϕ' is then corrected by $(N - M)/a$ the value of M will rapidly converge on N.

b. E and N from ϕ and λ

$$\text{I} = M + No \text{ (see paragraph a above)}$$

$$\text{II} = \frac{\nu}{2} \sin\phi \cos\phi$$

$$\text{III} = \frac{\nu}{24} \sin\phi \cos^3\phi(5 - \tan^2\phi + 9\eta^2)$$

$$\text{IIIA} = \frac{\nu}{720} \sin\phi \cos^5\phi(61 - 58\tan^2\phi + \tan^4\phi)$$

Then $N = (\text{I}) + P^2(\text{II}) + P^4(\text{III}) + P^6(\text{IIIA})$

$$\text{IV} = \nu \cos\phi$$

$$\text{V} = \frac{\nu}{6} \cos^3\phi\left(\frac{\nu}{\rho} - \tan^2\phi\right)$$

$$
\begin{aligned}
\text{VI} = \frac{\nu}{120}\cos^5\phi(&5 - 18\tan^2\phi + \tan^4\phi \\
&+ 14\eta^2 - 58\tan^2\phi\eta^2)
\end{aligned}
$$

Then $E = Eo + P(\text{IV}) + P^3(\text{V}) + P^5(\text{VI})$

c. ϕ and λ from E and N

$$\text{VII} = \frac{\tan \phi'}{2\rho\nu}$$

$$\text{VIII} = \frac{\tan \phi'}{24\rho\nu^3}(5 + 3\tan^2\phi' + \eta^2 - 9\tan^2\phi'\eta^2)$$

$$\text{IX} = \frac{\tan \phi'}{720\rho\nu^5}(61 + 90\tan^2\phi' + 45\tan^4\phi')$$

Then $\phi_p = \phi' - y^2(\text{VII}) + y^4(\text{VIII}) - y^6(\text{IX})$ (see paragraph 6a note 2)

$$\text{X} = \frac{\sec \phi'}{\nu}$$

$$\text{XI} = \frac{\sec \phi'}{6\nu^3}\left(\frac{\nu}{\rho} + 2\tan^2\phi'\right)$$

$$\text{XII} = \frac{\sec \phi'}{120\nu^5}(5 + 28\tan^2\phi' + 24\tan^4\phi')$$

$$\text{XIIA} = \frac{\sec \phi'}{5040\nu^7}(61 + 662\tan^2\phi' + 1320\tan^4\phi'$$

$$+ 720\tan^6\phi'$$

Then $\lambda_p = \lambda_0 + y(\text{X}) - y^3(\text{XI}) + y^5(\text{XII}) - y^7(\text{XIIA})$

d. C from ϕ and λ

$$\text{XIII} = \sin\phi$$

$$\text{XIV} = \frac{\sin\phi\cos^2\phi}{3}(1 + 3\eta^2 + 2\eta^4)$$

$$\text{XV} = \frac{\sin\phi\cos^4\phi}{15}(2 - \tan^2\phi)$$

Then $C = P(\text{XIII}) + P^3(\text{XIV}) + P^5(\text{VX})$

e. C from E and N

$$\text{XVI} = \frac{\tan\phi'}{\nu}$$

$$\text{XVII} = \frac{\tan\phi'}{3\nu^3}(1 + \tan^2\phi' - \eta^2 - 2\eta^4)$$

$$\text{XVIII} = \frac{\tan\phi'}{15\nu^5}(2 + 5\tan^2\phi' + 3\tan^4\phi')$$

(see paragraph 6a note 2)

Then $C = y(\text{XVI}) - y^3(\text{XVII}) + y^5(\text{XVIII})$

f. F from ϕ and λ

$$XIX = \frac{\cos^2 \phi}{2}(1 + \eta^2)$$

$$XX = \frac{\cos^4 \phi}{24}(5 - 4\tan^2 \phi + 14\eta^2 - 28\tan^2 \phi\eta^2)$$

Then $F = Fo(1 + P^2(XIX) + P^4(XX)$

g. F from E and N

$$XXI = \frac{1}{2\rho v}$$

$$XXII = \frac{(1 + 4\eta^2)}{24\rho^2 v^2} \quad (\eta, \rho, v \text{ derived from } \phi')$$

Then $F = Fo(1 + y^2(XXI) + y^4(XXII)$

h. $t - T$ from E and N

$$XXIII = \frac{1}{6\rho v}$$

1 and 2 are the terminals of the line

use $Nm = \frac{N_1 + N_2}{2}$ to compute ρ and v

Then $(t_1 - T_1) = (2y_1 + y_2)(N_1 - N_2)(XXIII)$

$(t_2 - T_2) = (2y_2 + y_1)(N_2 - N_1)(XXIII)$

7. Worked Examples

a. E, N from Latitude, Longitude

Caister Water Twr

Latitude	52 39 27.2531
Longitude	1 43 4.5177E
V	6.38850233E + 06
R	6.37275644E + 06
H2	2.47081362E − 03
M	4.06688296E + 05
P	6.48899730 − 02
I	3.06688296E + 05
II	1.54040791E + 06
III	1.560688E + 05
IIIA	−2.0671E + 04
IV	3.87512057E + 06
V	−1.700008E + 05
VI	−1.0134E + 05
Eastings	651 409.903
Northings	313 177.270

Framingham

Latitude	52 34 26.8915
Longitude	1 20 21.1080E
V	6.38847227E + 06
R	6.37266647E + 06
H2	2.48024893E − 03
M	3.97408391E + 05
P	5.82799762E − 02
I	2.97408391E + 05
II	1.54162270E + 06
III	1.572829E + 05
IIIA	−2.0516E + 04
IV	3.88249423E + 06
V	−1.685014E + 05
VI	−1.0165E + 05
Eastings	626 238.248
Northings	302 646.412

b. Latitude, Longitude from E, N

Caister Water Twr

Eastings	651 409.903
Northings	313 177.271
Phi'	52 42 57.2785
V	6.38852334E + 06
R	6.37281931E + 06
H2	2.46422052E − 03
M	4.13177271E + 05
y	2.51409903E + 05
VII	1.61305625E − 14
VIII	3.339555E − 28
IX	9.4199E − 42
X	2.58400625E − 07
XI	4.698597E − 21
XII	1.6124E − 34
XIIA	6.6577E − 48
Latitude	52 39 27.2531
Longitude	1 43 4.5177E

Framingham

Eastings	626 238.249
Northings	302 646.415
Phi'	52 37 16.4305
V	6.38848924E + 06
R	6.37271726E + 06
H2	2.47492225E − 03
M	4.02646415E + 05

y	2.26238249E + 05
VII	1.60757208E − 14
VIII	3.316668E − 28
IX	9.3107E − 42
X	2.57842725E − 07
XI	4.663699E − 21
XII	1.5922E − 34
XIIA	6.5408E − 48
Latitude	52 34 26.8916
Longitude	1 20 21.1081E

c. Convergence

The convergence C at any point in the projection is the angle between the "North-South" grid line and the direction of the meridian at that point. If the $(t - T)$ correction is neglected, then Grid Bearing + C = True Bearing. (But see para 7e). C is zero on the grid line $E = E_0$. It is positive to the east and negative to the west of this grid line. Remember that the meridians converge towards the North Pole which is situated on the E_0 grid line.

C from Latitude, Longitude

Framingham

Latitude	52 34 26.8915
Longitude	1 20 21.1080E
V	6.38847227E + 06
R	6.37266647E + 06
H2	2.48024893E − 03
XIII	7.941140369E − 01
XIV	9.849793 − 02
XV	2.1123E − 03
Convergence	2 39 10.4691

Caister Water Twr

Latitude	52 39 27.2531
Longitude	1 43 4.5177E
V	6.38850233E · 06
R	6.37275644E + 06
H2	2.47081362E − 03
XIII	7.95024300E − 01
XIV	9.822941E − 02
XV	2.0244E − 03
Convergence	2 57 26.5561

C from E, N

Framingham

Eastings	626 238.249
Northings	302 646.415
Phi'	52 37 16.4305
V	6.38848924E + 06
R	6.37271726E + 06
H2	2.47492225E − 03
XVI	2.04892047E − 07
XVII	4.536442E − 21
XVIII	1.588725E − 34
Convergence	2 39 10.4692

Caister Water Twr

Eastings	651 409.903
Northings	313 177.271
Phi'	52 42 57.2785
V	6.38852334E − 06
R	6.37281931E + 06
H2	2.46422052E − 03
XVI	2.05594320E − 07
XVII	4.571747E − 21
XVIII	1.608987E − 34
Convergence	2 57 26.5562

d. Scale Factors and True Distance from Rectangular Co-ordinates

In order to obtain the True Distance (S) from the Grid Distance (s) derived from Grid Co-ordinates; (or alternatively, in order to convert a true distance measured on the ground to a grid distance for plotting on the map or projection) it is necessary to calculate the Scale Factor and apply it in the correct sense.

The expression connecting these three quantities is

$$s = S \times F \quad \text{or} \quad S = s/F$$

The scale factor changes from point to point but so slowly that for most purposes it may be taken as constant within any 10 km square and equal to the value at the centre at the square considered. In the worst case the scale factor changes from one side of a 10 km square to the other by about 6 parts in 100 000. So that a value for the middle of the square would not be in error by more than 1/30 000 for any measurement made in that square.

For all practical purposes the Scale Factor may be taken as depending only on the distance from the central meridian. In the worst case the variation of scale from North to South of the projection along a line of constant easting is less than 1 in 600 000.

The table of scale factors given at the end of this pamphlet may be used for all ordinary work. Where greater accuracy is required the formula given in paragraph 6g may be used.

For a long line the factor should be calculated for the mid point of the line. For lines up to 30 km in length the mid point value will give results with an error not exceeding 1 or 2 parts in a million.

If still greater accuracy is needed compute a scale factor for both ends and the mid point and use Simpson's Rule, viz:

$$\frac{1}{F} = \frac{1}{6}\left(\frac{1}{F_1} + \frac{4}{F_m} + \frac{1}{F_2}\right)$$

F from Latitude, Longitude

Caister Water Twr

Latitude	52 39 27.2531
Longitude	1 43 4.5177E
V	6.38850233E + 06
R	6.37275644E + 06
H2	2.47081362E − 03
XIX	1.844226E − 01
XX	−1.1032E − 02
Local Scale	1.00037732

Framingham

Latitude	52 34 26.8915
Longitude	1 20 21.1080E
V	6.38847227E + 06
R	6.37266647E + 06
H2	2.48024893E − 03
XIX	1.851286E − 01
XX	−1.0879E − 02
Local Scale	1.00022970

F from E, N

Caister Water Twr

Eastings	651 409.903
Northings	313 177.271
Phi'	52 42 57.2785
V	6.38852334E + 06
R	6.37281931E + 06
H2	2.46422052E − 03
XXI	1.228112E − 14
XXII	2.5385E − 29
Local Scale	1.00037732

Framingham

Eastings	626 238.249
Northings	302 646.415
Phi'	52 37 16.4305
V	6.38848924E + 06
R	6.37271726E + 06
H2	2.47492225E − 03
XXI	1.228138E − 14
XXII	2.5388E − 29
Local Scale	1.00022969

Mid Point Framingham To Caister Water Twr

Eastings	638 824.076
Northings	307 911.843
Phi'	52 40 6.8552
V	6.38850630E + 06
R	6.37276830E + 06
H2	2.46957015E − 03
XXI	1.228125E − 14
XXII	2.5387E − 29
Local Scale	1.00030156

e. The Adjustment of Directions on the Projection or $(t - T)$ correction

The "$(t - T)$ correction" is the difference between the direction in nature and that on the projection.

The straight line joining the two points in nature (which, neglecting refraction, is practically identical with the geodesic on the spheroid) will normally be a curve when plotted on the Projection. The difference between the initial direction of that curve and the direction of the straight line joining the two points on the Projection is the $(t - T)$ correction.

In the diagram two plane triangles ABC are shown one on each side of the central meridian. The curved lines represent the geodesics or lines of sight. The curved geodesics are always concave towards the central meridian. AX and AY represent the tangents to the curves at A. The angles BAX and CAY represent the $(t - T)$ corrections at A to the lines AB and AC respectively. The sign of the correction for any given case is immediately clear from the diagram.

A similar diagram should always be drawn (with the aid of the rule underlined above) to help in applying the correction in the right sense. This method is less liable to lead to mistakes in sign than is a rule of thumb.

(The diagram may lead to confusion in the rare case of a line which crosses the Central Meridian. The $(t - T)$ correction is then bound to be very small and would only be considered in first order work of a very precise nature. In such cases a strict algebraic interpretation of the formulae is probably the safest rule).

Since the projection is conformal (i.e. directions at a point are maintained relatively correct) it follows that the True Bearing of B from A is given by the angle between the meridian at the tangent AX in the figure. But the Grid Bearing of B from A is the angle between Grid North and the line AB.

Therefore if the $(t - T)$ correction be taken into consideration,

True Bearing A → B = Grid Bearing A → B + C − $(t - T)$.



(t − T) from E, N

Framingham To Caister Water Twr

Eastings	626 238.249
Northings	302 646.415
Eastings	651 409.903
Northings	313 177.271
Phi'	52 40 6.8552
V	6.38850630E + 06
R	6.37276830E + 06
H2	2.46957015E − 03
XXIII	4.093750E − 15

(t − T)a (sec)	−6.26
(t − T)b (sec)	6.48

f. True Azimuth from Rectangular Co-ordinates

The True Azimuth is obtained by computing the grid bearing and applying the convergence and the $(t - T)$ correction. (Note that for short lines not exceeding 10 km in length the $(t - T)$ correction cannot exceed 7″ in the worst case on the limit of the projection. For minor surveys therefore it may be neglected).

The drawing of a rough diagram as shown over, from which the signs of C and $(t - T)$ can be seen by inspection is strongly recommended.



Station 1.	Framingham	E 626,238·249	N 302,646·415
Station 2.	Caister Water Twr.	E 651,409·903	N 313,177·271
		$E_2 - E_1$ + 25,171·654	$N_2 - N_1$ + 10,530·856

$$\frac{E_2 - E_1}{N_2 - N_1} = \text{tangent of plane grid bearing 1 to 2} = +2{\cdot}390\,276\,16$$

Plane grid bearing $(\alpha_1) = 67°\ 17'\ 50''\cdot759$

$$\sin \alpha_1 = 0{\cdot}922\,520\,80$$

$$\cos \alpha_1 = 0{\cdot}385\,947\,38$$

$$\text{Plane grid distance} = \frac{E_2 - E_1}{\sin \alpha_1} = 27{,}285{\cdot}730 = \frac{N_2 - N_1}{\cos \alpha_1} = 27{,}285{\cdot}730 \text{ (Check)}$$

$$\text{True Azimuth}_1 = \alpha_1 + C_1 - (t - T)''_1$$

$$\text{True Azimuth}_2 = \alpha_1 + C_2 - (t - T)''_2 + 180°$$

True Azimuth$_1$

α_1		67° 17′ 50″ ·759
$+C$	+	2° 39′ 10″ ·469
$-(t-T)''_1$	+	06″ ·259
True Azimuth$_1$	=	69° 57′ 07″ ·487

From previous examples.

True Azimuth$_2$

α_1		67° 17′ 50″ ·759
$+C_2$	+	2° 57′ 26″ ·556
$-(t-T)''_2$	−	06″ ·483
		+180° 00′ 00″ ·000
True Azimuth$_2$	=	250° 15′ 10″ ·832

For signs of C and $(t - T)$ see diagram above

Table of Local Scale Factor
(see paragraph 7d)

National Grid Easting (Km.)		Scale Factor *F*.
400	400	0·99960
390	410	60
380	420	61
370	430	61
360	440	62
350	450	63
340	460	65
330	470	66
320	480	68
310	490	70
300	500	72
290	510	75
280	520	78
270	530	81
260	540	84
250	550	88
240	560	92
230	570	0·99996
220	580	1·00000
210	590	04
200	600	09
190	610	14
180	620	20
170	630	25
160	640	31
150	650	37
140	660	43
130	670	1·00050

Use of Scale Factor

$s = S \times F$

Where s = distance in the projection
S = distance on the spheriod at mean sea level
F = Local Scale Factor from Table.

APPENDIX A.

```
10    ! Variables
20    ! Where possible a variable used in the following
30    ! routines is either the same as that used in the
40    ! formulae or can be deduced from the suffix.
50    !
60    ! No suffix = Upper Case Letter
70    ! Suffix 1 = Lower Case Letter
80    ! Suffix 2 = Lower Case Letter Squared
90    ! Suffix 0 = Subscript 0 e.g. E0 = Grid Eastings of True Origin.
100   !
110   ! J3—J9 are used for intermediate values.
120   !
130   ! The variables not covered by the above rules are listed below:
140   ! K = Phi (Latitude) or Phi'
150   ! L = Lambda (Longitude)
160   ! R = Rho (Radius of Curvature in Meridian)
170   ! V = Nu (Radius of Curvature in Prime Vertical)
180   ! H2 = Eta Squared (Nu/Rho - 1)
190   ! K3 = Phi2 - Phi1 (Difference Latitude)
200   ! K4 = Phi2 + Phi1 (Sum Latitudes)
210   ! Ga, Gb = (t - T) at line terminals A and B
220   ! All angular arguments are in Radians
230   !
240   !
250   ! Arc of Meridian
260   J3 = (1 + N1 + 5/4*N1^2 + 5/4*N1^3)*K3
270   J4 = (3*N1 + 3*N1^2 + 21/8*N1^3)*SIN(K3)*COS(K4)
280   J5 = (15/8*N1^2 + 15/8*N1^3)*SIN(2*K3)*COS(2*K4)
290   J6 = 35/24*N1^3*SIN(3*K3)*COS(3*K4)
300   M = B1*(J3 - J4 + J5 - J6)
310   RETURN
320   !
330   !
340   ! Compute Phi' (K)
350   K = (N - N0)/A1 + K0
360   K3 = K - K0
370   K4 = K + K0
380   GOSUB 260
390   IF ABS(N - N0 - M) < .001 THEN 420
400   K = K + (N - N0 - M)/A1
410   GOTO 360
420   RETURN
430   !
440   !
450   ! Compute V, R&H2
460   V = A1/SQR(1 - E2*SIN(K)^2)
470   R = V*(1 - E2)/(1 - E2*SIN(K)^2)
480   H2 = V/R - 1
490   RETURN
```

```
500  !
510  !
520  !  E&N from Latitude (K) & Longitude (L)
530  K3 = K - K0
540  K4 = K + K0
550  GOSUB 260
560  GOSUB 460
570  P = L - L0
580  J3 = M + N0                                                          !  I
590  J4 = V/2*SIN(K)*COS(K)                                               !  II
600  J5 = V/24*SIN(K)*COS(K)^3*(5 - TAN(K)^2 + 9*H2)                      !  III
610  J6 = V/720*SIN(K)*COS(K)^5*(61 - 58*TAN(K)^2 + TAN(K)^4)             !  IIIA
620  N = J3 + P^2*J4 + P^4*J5 + P^6*J6
630  J7 = V*COS(K)                                                        !  IV
640  J8 = V/6*COS(K)^3*(V/R - TAN(K)^2)                                   !  V
650  J9 = V/120*COS(K)^5
660  J9 = J9*(5 - 18*TAN(K)^2 + TAN(K)^4 + 14*H2 - 58*TAN(K)^2*H2)        !  VI
670  E = E0 + P*J7 + P^3*J8 + P^5*J9
680  RETURN
690  !
700  !
710  !  Latitude & Longitude from E & N
720  GOSUB 350
730  GOSUB 460
740  Y1 = E - E0
750  J3 = TAN(K)/(2*R*V)                                                  !  VII
760  J4 = TAN(K)/(24*R*V^3)*(5 + 3*TAN(K)^2 + H2 - 9*TAN(K)^2*H2)         !  VIII
770  J5 = TAN(K)/(720*R*V^5)*(61 + 90*TAN(K)^2 + 45*TAN(K)^4)             !  IX
780  K9 = K - Y1^2*J3 + Y1^4*J4 - Y1^6*J5
790  J6 = 1/(COS(K)*V)                                                    !  X
800  J7 = 1/(COS(K)*6*V^3)*(V/R + 2*TAN(K)^2)                             !  XI
810  J8 = 1/(COS(K)*120*V^5)*(5 + 28*TAN(K)^2 + 24*TAN(K)^4)              !  XII
820  J9 = 1/(COS(K)*5040*V^7)
830  J9 = J9*(61 + 662*TAN(K)^2 + 1320*TAN(K)^4 + 720*TAN(K)^6)           !  XIIA
840  L = L0 + Y1*J6 - Y1^3*J7 + Y1^5*J8 - Y1^7*J9
850  K = K9
860  RETURN
870  !
880  !
890  !  C from Latitude & Longitude
900  GOSUB 460
910  P = L - L0
920  J3 = SIN(K)                                                          !  XIII
930  J4 = SIN(K)*COS(K)^2/3*(1 + 3*H2 + 2*H2^2)                           !  XIV
940  J5 = SIN(K)*COS(K)^4/15*(2 - TAN(K)^2)                               !  XV
950  C = P*J3 + P^3*J4 + P^5*J5
960  RETURN
970  !
```

```
980   !
990.  !  C from E & N
1000  GOSUB 350
1010  GOSUB 460
1020  Y1 = E - E0
1030  J3 = TAN(K)/V                                                  ! XVI
1040  J4 = TAN(K)/(3*V^3)*(1 + TAN(K)^2 - H2 - 2*H2^2)               ! XVII
1050  J5 = TAN(K)/(15*V^5)*(2 + 5*TAN(K)^2 + 3*TAN(K)^4)             ! XVIII
1060  C = Y1*J3 - Y1^3*J4 + Y1^5*J5
1070  RETURN
1080  !
1090  !
1100  !  F from Latitude & Longitude
1110  GOSUB 460

1120  P = L - L0
1130  J3 = COS(K)^2/2*(1 + H2)                                       ! XIX
1140  J4 = COS(K)^4/24*(5 - 4*TAN(K)^2 + 14*H2 - 28*TAN(K)^2*H2)     ! XX
1150  F = F0*(1 + P^2*J3 + P^4*J4)
1160  RETURN
1170  !
1180  !
1190  !  F from E & N
1200  GOSUB 350
1210  GOSUB 460
1220  Y1 = E - E0

1230  J3 = 1/(2*R*V)                                                 ! XXI
1240  J4 = (1 + 4*H2)/(24*R^2*V^2)                                   ! XXII
1250  F = F0*(1 + Y1^2*J3 + Y1^4*J4)
1260  RETURN
1270  !
1280  !
1290  !  (t - T) from E, N
1300  N = (Na + Nb)/2
1310  GOSUB 350
1320  GOSUB 460
1330  J3 = 1/(6*R*V)                                                 ! XXIII
1340  Ga = (2*Y1a + Y1b)*(Na - Nb)*J3
1350  Gb = (2*Y1b + Y1a)*(Nb - Na)*J3
1360  RETURN
```